Multiway ANOVA

NESTED DESIGN

Nested Designs

Crossed vs. Nested Design

Heuristic Definition: "Crossed is all combinations, nested is more limited"

Are explanatory variables **CROSSED** or **NESTED**?

Formal Definition: If the levels of one of the factors occur at all of the levels of the other factors, then they are **CROSSED** (sometimes referred to as **FACTORIAL** design).

A factor B is **NESTED** in a factor A if each level of B occurs in only one level of A

Example: Suppose we work for 3M and we manufacture abrasives at 3 sites. We want to examine the durability of these abrasives. Due to particulars of the manufacturing process, two different abrasives can be produced at each site. We also believe that the each site is an effect of interest.

Here, abrasives is nested in site, which is notated abrasive(site)

Further Specifics of this Experiment

We will use an abrasive until it deteriorates past some threshold

Each abrasive will be tested twice

As there are 6 abrasives and 2 replications, we have 12 total trials

However, as the design is nested, there isn't the same analysis as with a cross design

Note that site is a bit like a block. However, there are two important differences:

- We are interested in the site effect
- All levels of the treatment don't occur inside each site

Abrasive Data

```
DATA abrasive1;
    INPUT site $ abrasive times;
    DATALINES;
    site1 1 25
    site1 1 29
    site1 2 14
    site1 2 11
    site2 1 11
    site2 1 6
    site2 2 22
    site2 2 18
    site3 1 17
    site3 1 20
    site3 2 5
    site3 2 2
    ;
RUN;
```

Important: abrasive = 1 doesn't have the same meaning for different sites

Nested Model

Instead of the ANOVA representation for a nested, interaction model:

$Y_{ijk} = \mu + \mu_j + \mu_k + \mu_{jk} + \varepsilon_{ijk}$

We introduce the nesting operator: $k(j)$

$Y_{ijk} = \mu + \mu_j + \mu_{k(j)} + \varepsilon_{ijk}$

This encodes the idea that $k = 1,2$ for abrasive 1 or 2, but abrasive 1 at site 1 is not the same as abrasive 1 at site 2

In this notation, $k(1) = 1$ is not the same as $k(2) = 1$, even though both equal 1

Test for site. H_0: all $\mu_j = 0$

Test for abrasive. H_0: all $\mu_{k(j)} = 0$

Nested Model: The Design Matrices

Crossed/Interaction

Nested

```
PROC GLMMOD DATA = abrasive1;
    CLASS site abrasive;
    MODEL times = site abrasive site*abrasive;
RUN;
```

```
PROC GLMMOD DATA = abrasive1;
    CLASS site abrasive;
    MODEL times = site abrasive(site);
RUN;
```

Parameter Definitions

Column Number	Name of Associated Effect	CLASS Variable Values	
		site	abrasive
1	Intercept		
2	site	site1	
3	site	site2	
4	site	site3	
5	abrasive		1
6	abrasive		2
7	site*abrasive	site1	1
8	site*abrasive	site1	2
9	site*abrasive	site2	1
10	site*abrasive	site2	2
11	site*abrasive	site3	1
12	site*abrasive	site3	2

Parameter Definitions

Column Number	Name of Associated Effect	CLASS Variable Values	
		site	abrasive
1	Intercept		
2	site	site1	
3	site	site2	
4	site	site3	
5	abrasive(site)	site1	1
6	abrasive(site)	site1	2
7	abrasive(site)	site2	1
8	abrasive(site)	site2	2
9	abrasive(site)	site3	1
10	abrasive(site)	site3	2

Nested Model: The Design Matrices

Crossed/Interaction

```
PROC GLMMOD DATA = abrasive1;
    CLASS site abrasive;
    MODEL times = site abrasive site*abrasive;
RUN;
```

Nested

```
PROC GLMMOD DATA = abrasive1;
    CLASS site abrasive;
    MODEL times = site abrasive(site);
RUN;
```

Design Points													
		Column Number											
Observation Number	times	1	2	3	4	5	6	7	8	9	10	11	12
1	25	1	1	0	0	1	0	1	0	0	0	0	0
2	29	1	1	0	0	1	0	1	0	0	0	0	0
3	14	1	1	0	0	0	1	0	1	0	0	0	0
4	11	1	1	0	0	0	1	0	1	0	0	0	0
5	11	1	0	1	0	1	0	0	0	1	0	0	0
6	6	1	0	1	0	1	0	0	0	1	0	0	0
7	22	1	0	1	0	0	1	0	0	0	1	0	0
8	18	1	0	1	0	0	1	0	0	0	1	0	0
9	17	1	0	0	1	1	0	0	0	0	0	1	0
10	20	1	0	0	1	1	0	0	0	0	0	1	0
11	5	1	0	0	1	0	1	0	0	0	0	0	1
12	2	1	0	0	1	0	1	0	0	0	0	0	1

Design Points											
		Column Number									
Observation Number	times	1	2	3	4	5	6	7	8	9	10
1	25	1	1	0	0	1	0	0	0	0	0
2	29	1	1	0	0	1	0	0	0	0	0
3	14	1	1	0	0	0	1	0	0	0	0
4	11	1	1	0	0	0	1	0	0	0	0
5	11	1	0	1	0	0	0	1	0	0	0
6	6	1	0	1	0	0	0	1	0	0	0
7	22	1	0	1	0	0	0	0	1	0	0
8	18	1	0	1	0	0	0	0	1	0	0
9	17	1	0	0	1	0	0	0	0	1	0
10	20	1	0	0	1	0	0	0	0	1	0
11	5	1	0	0	1	0	0	0	0	0	1
12	2	1	0	0	1	0	0	0	0	0	1

An Improper Analysis

```
PROC GLMMOD DATA = abrasive1;
    CLASS site abrasive;
    MODEL times = site abrasive site*abrasive;
RUN;
```

Source	DF	Type III SS	Mean Square	F Value	Pr > F
site	2	156.5000000	78.2500000	11.18	0.0095
abrasive	1	108.0000000	108.0000000	15.43	0.0077
site*abrasive	2	459.5000000	229.7500000	32.82	0.0006

SAS will happily run the analysis as a crossed design

However, this won't give you the right results as this presumes the same abrasive has been tested 6 times in 3 block

A Proper Analysis

```
PROC GLM DATA = abrasive1 PLOTS = all;
    CLASS site abrasive;
    MODEL times = site abrasive(site);
RUN;
```

Source	DF	Type III SS	Mean Square	F Value	Pr > F
site	2	156.5000000	78.2500000	11.18	0.0095
abrasive(site)	3	567.5000000	189.1666667	27.02	0.0007



These are typical diagnostic plots from a multi-way ANOVA

Due to the discrete nature and small sample size, you should just look for egregious violations (Here, these look fine)

There is evidence to suggest that there is a difference in mean for both the sites and the abrasives.

Test for site. H_0: all $\mu_j = 0$

Test for abrasive. H_0: all $\mu_{k(j)} = 0$

(reject both hypotheses at 0.05 level)

Which Means are Different? Looking at Sites

```
PROC GLM DATA = abrasive1 PLOTS = all;
    CLASS site abrasive;
    MODEL times = site abrasive(site);
    LSMEANS site abrasive(site) / PDIFF CL;
RUN;
```

The GLM Procedure
Least Squares Means

site	times LSMEAN	LSMEAN Number
site1	19.7500000	1
site2	14.2500000	2
site3	11.0000000	3

Least Squares Means for effect site
Pr > |t| for H0: LSMean(i)=LSMean(j)

Dependent Variable: times

i/j	1	2	3
1		0.0260	0.0034
2	0.0260		0.1330
3	0.0034	0.1330	

Don't forget to correct for multiple comparisons…

```
PROC GLM DATA = abrasive1 PLOTS = all;
    CLASS site abrasive;
    MODEL times = site abrasive(site);
    LSMEANS site abrasive(site) / PDIFF CL ADJUST=tukey;
RUN;
```

The GLM Procedure
Least Squares Means
Adjustment for Multiple Comparisons: Tukey

site	times LSMEAN	LSMEAN Number
site1	19.7500000	1
site2	14.2500000	2
site3	11.0000000	3

Least Squares Means for effect site
Pr > |t| for H0: LSMean(i)=LSMean(j)

Dependent Variable: times

i/j	1	2	3
1		0.0586	0.0081
2	0.0586		0.2677
3	0.0081	0.2677	

Which Means are Different? Looking at Sites

Least Squares Means for Effect site				
i	j	Difference Between Means	Simultaneous 95% Confidence Limits for LSMean(i)-LSMean(j)	
1	2	5.500000	-0.240181	11.240181
1	3	8.750000	3.009819	14.490181
2	3	3.250000	-2.490181	8.990181

There is evidence that the means of site 1 and site 3 are different. We estimate a 8.75 s difference, (holding abrasive type constant) with a range of plausible values for this difference of 3s to 14.5s



Least Squares Means for effect site
Pr > |t| for H0: LSMean(i)=LSMean(j)

Dependent Variable: times

i/j	1	2	3
1		0.0586	0.0081
2	0.0586		0.2677
3	$\boxed{0.0081}$	0.2677	

Which Means are Different? Looking at Abrasives

Least Squares Means for Effect abrasive(site)				
i	j	Difference Between Means	Simultaneous 95% Confidence Limits for LSMean(i)-LSMean(j)	
1	2	14.500000	3.970694	25.029306
1	3	18.500000	7.970694	29.029306
1	4	7.000000	-3.529306	17.529306
1	5	8.500000	-2.029306	19.029306
1	6	23.500000	12.970694	34.029306
2	3	4.000000	-6.529306	14.529306
2	4	-7.500000	-18.029306	3.029306
2	5	-6.000000	-16.529306	4.529306
2	6	9.000000	-1.529306	19.529306
3	4	-11.500000	-22.029306	-0.970694
3	5	-10.000000	-20.529306	0.529306
3	6	5.000000	-5.529306	15.529306
4	5	1.500000	-9.029306	12.029306
4	6	16.500000	5.970694	27.029306
5	6	15.000000	4.470694	25.529306

Least
F

i/j	1	2	3	4	5	6
1		0.0115	0.0033	0.2179	0.1155	0.0009
2	0.0115		0.6712	0.1764	0.3294	0.0936
3	0.0033	0.6712		0.0341	0.0619	0.4836
4	0.2179	0.1764	0.0341		0.9899	0.0060
5	0.1155	0.3294	0.0619	0.9899		0.0097
6	0.0009	0.0936	0.4836	0.0060	0.0097	



Nested vs. Crossed/Additive Model

Nested vs. Crossed/Additive Model

```
DATA abrasive1;
    INPUT site $ abrasive times;
    DATALINES;
    site1 1 25
    site1 1 29
    site1 2 14
    site1 2 11
    site2 1 11
    site2 1 6
    site2 2 22
    site2 2 18
    site3 1 17
    site3 1 20
    site3 2 5
    site3 2 2
    ;
RUN;
```

Original data

```
DATA abrasive2;
    INPUT site $ abrasive times;
    DATALINES;
    site1 1 25
    site1 1 29
    site1 2 14
    site1 2 11
    site2 3 11
    site2 3 6
    site2 4 22
    site2 4 18
    site3 5 17
    site3 5 20
    site3 6 5
    site3 6 2
    ;
RUN;
```

Coding abrasive as a unique value

Nested vs. Crossed/Additive Model

```
PROC GLM DATA = abrasive1 PLOTS = all;
    CLASS site abrasive;
    MODEL times = site abrasive(site);
RUN;
```

Measures variability of…

- **each abrasive with respect to its site**
- each site with respect to the overall average

Experimental Unit for Abrasive: Test Run (yes replication)
Experimental Unit for Site: Batch of 4 test runs (no replication)
(hence this is an example of a split plot)

```
PROC GLM DATA = abrasive2 PLOTS = all;
    CLASS site abrasive;
    MODEL times = site abrasive;
RUN;
```

Measures variability of…

- **each abrasive with respect to the overall average**
- each site with respect to the overall average

Nested vs. Crossed/Additive Model

Nested: We get an estimate of the error by assessing the variability of abrasive around each site mean

Measures variability of…

- **each abrasive with respect to its site**
- each site with respect to the overall average

Experimental Unit for Abrasive: Test Run (yes replication)
Experimental Unit for Site: Batch of 4 test runs (no replication)
(hence this is an example of a split plot)

Crossed/Additive: We cannot get an estimate of the error due to no replication for site

Measures variability of…

- **each abrasive with respect to the overall average**
- each site with respect to the overall average

Nested vs. Crossed/Additive Model

```
PROC GLM DATA = abrasive1 PLOTS = all;
    CLASS site abrasive;
    MODEL times = site abrasive(site);
RUN;
```

Source	DF	Type III SS	Mean Square	F Value	Pr > F
site	2	156.5000000	78.2500000	11.18	0.0095
abrasive(site)	3	567.5000000	189.1666667	27.02	0.0007

Source	DF	Type III SS	Mean Square	F Value	Pr > F
site	0	0.0000000	.	.	.
abrasive	3	567.5000000	189.1666667	27.02	0.0007

```
PROC GLM DATA = abrasive2 PLOTS = all;
    CLASS site abrasive;
    MODEL times = site abrasive;
RUN;
```

Split-plot Design

Split-plot

Back to Example: We have a field of 8 plots allocated for testing two different species of corn. Irrigation timing has an affect that we would like to control for. We have to irrigate groups of 4 plots at the same time and assign treatments to an entire plot

So we have an $\mathrm{A} =$ irrigation, $\mathrm{B} =$ treatment

What are the experiment units?

- irrigation: group of 4 plots (known as the **WHOLE PLOT**)
- treatment: individual plot (known as the **SPLIT PLOT**)

Often, you will get combinations of split-plots and cross factors

Lawnmower Example

Suppose we have two manufacturers of lawnmowers

Each manufacturer produces 3 types of lawnmowers

We run each lawnmower at a “high” and a “low” setting

We record the amount of time it takes to stop the blade once power is cut (for safety) twice for each “high” and “low”

We want to compare these “cut-off” times for the lawnmowers

Testing for Interaction for Crossed Effect

```
PROC GLM DATA = mower PLOTS = all;
    CLASS manufac mower speed;
    MODEL times = manufac speed mower(manufac) manufac*speed;
RUN;
```

Source	DF	Type III SS	Mean Square	F Value	Pr > F
manufac	1	2521.50000	2521.50000	19.06	0.0005
speed	1	20886.00000	20886.00000	157.87	<.0001
mower(manufac)	4	2852.83333	713.20833	5.39	0.0061
manufac*speed	1	10.66667	10.66667	0.08	0.7801



Fitted Combined Additive, Nested Model

```
PROC GLM DATA = mower PLOTS = all;
    CLASS manufac mower speed;
    MODEL times = manufac speed mower(manufac);
    LSMEANS manufac speed mower(manufac)/ PDIFF CL ADJUST=tukey;
RUN;
```

Source	DF	Type III SS	Mean Square	F Value	Pr > F
manufac	1	2521.50000	2521.50000	20.15	0.0003
speed	1	20886.00000	20886.00000	166.89	<.0001
mower(manufac)	4	2852.83333	713.20833	5.70	0.0043



Fitted Combined Additive, Nested Model